
Change 1
	Function name
	f_IMS_MTCall_Setup_SendPRACK_ReceiveOK()

	Reason for change
	In the codec selection test cases 16.3 and 16.4, the UE can send 180 Ringing immediately following bearer activation. This means it could be received while the TTCN is waiting for 200 OK for PRACK.

	Summary of change
	Add a new boolean paramater “AllowRinging”. If this paramter is true, handle any 180 Ringing response received while waiting for 200 OK for PRACK. If received, return the contents of the 180 response from the function. 

	TTCN module
	IMS_Proedures_CallControl.ttcn


Before change

  function f_IMS_MTCallSetup_SendPRACK_ReceiveOK(template (value) INVITE_Request p_InviteRequest,

                                                 MessageHeader p_MessageHeader_PrevReliableResponse,

                                                 template (omit) ContentType p_ContentType := omit) runs on IMS_PTC

  {

    var template (value) MessageHeader v_MessageHeader_Prack;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    // Send PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, p_MessageHeader_PrevReliableResponse, p_ContentType);

    IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

    // Receive 200 OK

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

  }
After change

  function f_IMS_MTCallSetup_SendPRACK_ReceiveOK(template (value) INVITE_Request p_InviteRequest,

                                                 MessageHeader p_MessageHeader_PrevReliableResponse,

                                                 template (omit) ContentType p_ContentType := omit,

                                                 boolean p_AllowRinging := false) runs on IMS_PTC return template (omit) MessageHeader
  {

    var template (value) MessageHeader v_MessageHeader_Prack;

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var template (omit) MessageHeader v_MessageHeader180 := omit;

    var template (omit) IMS_DATA_RSP v_IMS_DATA_RSP;
    // Send PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderTX(p_InviteRequest, p_MessageHeader_PrevReliableResponse, p_ContentType);

    IMS_Client.send(cas_IMS_Prack_Request(cs_IMS_RoutingInfo(v_Protocol), cs_PRACK_Request(v_ContactUrl, v_MessageHeader_Prack)));

    // Receive 200 OK and optionally 180 ringing
    alt{
        [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Prack))));

      [p_AllowRinging] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest))))

        -> value v_IMS_DATA_RSP

        {

            v_MessageHeader180 := v_IMS_DATA_RSP.Response.msgHeader;

            repeat;

        }

    }

    return v_MessageHeader180;
  }
Change 2
	Function name
	f_IMS_MTCall_Setup_Common_Steps7_11()

	Reason for change
	Following change 1 above, the 180 Ringing response from the UE may have already been received before this function is called.

	Summary of change
	Add a new parameter to pass in the contents of any previously received 180 response to this function.If a 180 Response is passed into the function, the function will not wait to receive the 180 and will generate the PRACK based on the message content that was passed as a parameter.

	TTCN module
	IMS_Proedures_CallControl.ttcn


Before change

  function f_IMS_MTCallSetup_Common_Steps7_11(template (value) INVITE_Request p_InviteRequest,

                                              template (value) SDP_Message p_SDP_MessageStep7,

                                              template (present) SDP_Message p_SDP_MessageStep8,

                                              InternetProtocol_Type p_ProtocolUsedForInvite,

                                              float p_WaitDuration_Ringing := 5.0) runs on IMS_PTC return template (omit) MessageHeader

  { /* @sic R5-140975: p_MessageBodyRinging removed as it is never used i.e. always omit sic@ */

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Update;

    var MessageHeader v_MessageHeader_Ringing;

    var InternetProtocol_Type v_ProtocolForUpdate := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var boolean v_WaitingFor180 := true;

    var boolean v_WaitingFor200 := true;

    var template ContentType v_ContentTypeRinging := omit;                                /* @sic R5-140975 sic@ */

    var template (present) ContentLength v_ContentLengthRinging := cr_ContentLength(0);   /* @sic R5-140975 sic@ */

    var template MessageBody v_MessageBodyRinging := omit;                                /* @sic R5-140975 sic@ */

    timer t_Timer := p_WaitDuration_Ringing;

    // Send UPDATE

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest);

    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_ProtocolForUpdate), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(p_SDP_MessageStep7))));

    t_Timer.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    while (v_WaitingFor180 or v_WaitingFor200) {

      alt {

        // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

        [v_WaitingFor200] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtocolForUpdate),

                                                              cr_Response(c_statusLine200,

                                                                          f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cs_ContentTypeSDP), /* @sic R5s130685 change 4 sic@ */

                                                                          cr_MessageBody_SDP(p_SDP_MessageStep8))))

          -> value v_IMS_DATA_RSP

          {

            // check at least one connections field sent

            f_IMS_SDP_CheckConnections(v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

            v_WaitingFor200 := false;

          }

        [v_WaitingFor180] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(p_ProtocolUsedForInvite),

                                                              cr_Response(c_statusLine180,

                                                                          f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest, -, v_ContentTypeRinging, v_ContentLengthRinging),

                                                                          v_MessageBodyRinging)))

          -> value v_IMS_DATA_RSP

          {

            t_Timer.stop;

            v_MessageHeader_Ringing := v_IMS_DATA_RSP.Response.msgHeader;

            v_WaitingFor180 := false;

          }

        [] t_Timer.timeout

          {

            v_WaitingFor180 := false;

          }

      }

    }

    if (isbound(v_MessageHeader_Ringing)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_Ringing);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader_Ringing, c_tag100rel)) {

        // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(p_InviteRequest, v_MessageHeader_Ringing);

      }

      return v_MessageHeader_Ringing;

    }

    return omit;

  }
After change

    function f_IMS_MTCallSetup_Common_Steps7_11(template (value) INVITE_Request p_InviteRequest,

                                              template (value) SDP_Message p_SDP_MessageStep7,

                                              template (present) SDP_Message p_SDP_MessageStep8,

                                              InternetProtocol_Type p_ProtocolUsedForInvite,

                                              float p_WaitDuration_Ringing := 5.0,

                                              template (omit) MessageHeader p_MessageHeader180 := omit) runs on IMS_PTC return template (omit) MessageHeader

  { /* @sic R5-140975: p_MessageBodyRinging removed as it is never used i.e. always omit sic@ */

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Update;

    var MessageHeader v_MessageHeader_Ringing;

    var InternetProtocol_Type v_ProtocolForUpdate := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var boolean v_WaitingFor180 := true;

    var boolean v_WaitingFor200 := true;

    var template ContentType v_ContentTypeRinging := omit;                                /* @sic R5-140975 sic@ */

    var template (present) ContentLength v_ContentLengthRinging := cr_ContentLength(0);   /* @sic R5-140975 sic@ */

    var template MessageBody v_MessageBodyRinging := omit;                                /* @sic R5-140975 sic@ */

    timer t_Timer := p_WaitDuration_Ringing;

    //180 ringing passed into function

    if (isvalue (p_MessageHeader180))

    {

        v_MessageHeader_Ringing := p_MessageHeader180;

        v_WaitingFor180 := false;

    }
    // Send UPDATE

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest);

    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_ProtocolForUpdate), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(p_SDP_MessageStep7))));

    t_Timer.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    while (v_WaitingFor180 or v_WaitingFor200) {

      alt {

        // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

        [v_WaitingFor200] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtocolForUpdate),

                                                              cr_Response(c_statusLine200,

                                                                          f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cs_ContentTypeSDP), /* @sic R5s130685 change 4 sic@ */

                                                                          cr_MessageBody_SDP(p_SDP_MessageStep8))))

          -> value v_IMS_DATA_RSP

          {

            // check at least one connections field sent

            f_IMS_SDP_CheckConnections(v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

            v_WaitingFor200 := false;

          }

        [v_WaitingFor180] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(p_ProtocolUsedForInvite),

                                                              cr_Response(c_statusLine180,

                                                                          f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest, -, v_ContentTypeRinging, v_ContentLengthRinging),

                                                                          v_MessageBodyRinging)))

          -> value v_IMS_DATA_RSP

          {

            t_Timer.stop;

            v_MessageHeader_Ringing := v_IMS_DATA_RSP.Response.msgHeader;

            v_WaitingFor180 := false;

          }

        [] t_Timer.timeout

          {

            v_WaitingFor180 := false;

          }

      }

    }

    if (isbound(v_MessageHeader_Ringing)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_Ringing);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader_Ringing, c_tag100rel)) {

        // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(p_InviteRequest, v_MessageHeader_Ringing);

      }

      return v_MessageHeader_Ringing;

    }

    return omit;

  }

Change 3
	Function name
	f_TC_16_3_and_16_4_TestBody()

	Reason for change
	At step 1 of test case 16.3 and 16.4, the SS sends an INVITE including the SDP attribute “a=curr:qos local sendrecv” which indicates to the UE that resources have already been reserved on the originating side of the call. This means that the UE may alert the user to the incoming call as soon as its local resource reservation is complete. Due to this the UE can send the “180 ringing” message at any point following the completion of dedicated EPS bearer context activation.

	Summary of change
	Set new parameter “AllowRinging” to TRUE when calling f_IMS_MTCallSetup_SendPRACK_ReceiveOK() (see change 1). Save any 180 Ringing response returned from this function and pass into call to f_IMS_MTCallSetup_Common_Steps7_11() (see change 2)

	TTCN module
	IMS_CC_CodecSelecting.ttcn


Before change

    function f_TC_16_3_and_16_4_TestBody(CharStringList_Type p_MediaFormat1ParametersTX,

                                       CharStringList_Type p_MediaFormat2ParametersTX,

                                       CharStringList_Type p_MediaFormat1ParametersRX) runs on IMS_PTC

  {

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var Response v_Response;

    var MessageHeader v_MessageHeader;

    var SDP_Message v_SDP_Message_Step4or9or12;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (present) MessageHeader v_MessageHeader_183RX;

    var template (value) SDP_Message v_SDP_Message_TX;

    var template (present) SDP_Message v_SDP_Message_RX;

    var template SDP_Message v_SDP_Message_Step12 := omit;

    var boolean v_Trying := false;

    var charstring v_Direction;

    var charstring v_Fmt1 := "97";

    var charstring v_Fmt2 := "99";

.........
          // Step 5 - 6. Send PRACK, Receive 200 OK

          f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader);

          // Step 7 - 11. Send UPDATE, Receive 200 OK and optional 180 Ringing

          // get curr:qos remote for SDP message in step 7

          if (ispresent(v_SDP_Message_Step4or9or12.media_list[0]) and f_SDP_AttributeList_CheckAttribute(v_SDP_Message_Step4or9or12.media_list[0].attributes, cr_SDP_Attribute_curr_quos(c_local, c_none))) {  // see Note 1

            v_Direction := c_none;

          } else {

            v_Direction := c_sendrecv;

          }

          v_SDP_Message_TX := f_IMS_BuildSDP_1AMR_WB_TX(v_Fmt1, p_MediaFormat1ParametersTX, v_Direction, c_mandatory);

          v_SDP_Message_RX := f_IMS_BuildSDP_TC_16_AMR_WB_RX(v_Fmt1, p_MediaFormat1ParametersRX, c_sendrecv, c_sendrecv);

          f_IMS_MTCallSetup_Common_Steps7_11(v_InviteRequest, v_SDP_Message_TX, v_SDP_Message_RX, v_Protocol);

        }

      // Step 9 - Receive optional 180 Ringing

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(v_InviteRequest), cr_MessageBody_SDP(v_SDP_Message_RX) ifpresent)))

        -> value v_IMS_DATA_RSP

        {

          t_Timer.stop;

          v_Response := v_IMS_DATA_RSP.Response;

          v_MessageHeader := v_IMS_DATA_RSP.Response.msgHeader;

          f_IMS_Dialog_SetRemoteTag(v_MessageHeader);

          if (ischosen(v_Response.messageBody.sdpMessageBody)) {

            v_SDP_Message_Step4or9or12 := v_Response.messageBody.sdpMessageBody;

          }

          f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

          if (f_MessageHeader_CheckRequire(v_MessageHeader, c_tag100rel)) {

            // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

            f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader);

          }

          f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer   @sic R5s130956 Change 2 sic@

        }

      [] t_Timer.timeout

        {

          f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer   @sic R5s130956 Change 2 sic@

        }

    }

    if (not isbound(v_SDP_Message_Step4or9or12)) {

      v_SDP_Message_Step12 := v_SDP_Message_RX;

    }

.......
  }

After change

      function f_TC_16_3_and_16_4_TestBody(CharStringList_Type p_MediaFormat1ParametersTX,

                                       CharStringList_Type p_MediaFormat2ParametersTX,

                                       CharStringList_Type p_MediaFormat1ParametersRX) runs on IMS_PTC

  {

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var Response v_Response;

    var MessageHeader v_MessageHeader;

    var SDP_Message v_SDP_Message_Step4or9or12;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (present) MessageHeader v_MessageHeader_183RX, 

    var template (omit) v_MessageHeader180;
    var template (value) SDP_Message v_SDP_Message_TX;

    var template (present) SDP_Message v_SDP_Message_RX;

    var template SDP_Message v_SDP_Message_Step12 := omit;

    var boolean v_Trying := false;

    var charstring v_Direction;

    var charstring v_Fmt1 := "97";

    var charstring v_Fmt2 := "99";

    ..........
          // Step 5 - 6. Send PRACK, Receive 200 OK (and optionally 180 Ringing)
          v_MessageHeader180 := f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader, -, true);
          // Step 7 - 11. Send UPDATE, Receive 200 OK and optional 180 Ringing

          // get curr:qos remote for SDP message in step 7

          if (ispresent(v_SDP_Message_Step4or9or12.media_list[0]) and f_SDP_AttributeList_CheckAttribute(v_SDP_Message_Step4or9or12.media_list[0].attributes, cr_SDP_Attribute_curr_quos(c_local, c_none))) {  // see Note 1

            v_Direction := c_none;

          } else {

            v_Direction := c_sendrecv;

          }

          v_SDP_Message_TX := f_IMS_BuildSDP_1AMR_WB_TX(v_Fmt1, p_MediaFormat1ParametersTX, v_Direction, c_mandatory);

          v_SDP_Message_RX := f_IMS_BuildSDP_TC_16_AMR_WB_RX(v_Fmt1, p_MediaFormat1ParametersRX, c_sendrecv, c_sendrecv);

          f_IMS_MTCallSetup_Common_Steps7_11(v_InviteRequest, v_SDP_Message_TX, v_SDP_Message_RX, v_Protocol, -, v_MessageHeader180);
        }

      // Step 9 - Receive optional 180 Ringing

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(v_InviteRequest), cr_MessageBody_SDP(v_SDP_Message_RX) ifpresent)))

        -> value v_IMS_DATA_RSP

        {

          t_Timer.stop;

          v_Response := v_IMS_DATA_RSP.Response;

          v_MessageHeader := v_IMS_DATA_RSP.Response.msgHeader;

          f_IMS_Dialog_SetRemoteTag(v_MessageHeader);

          if (ischosen(v_Response.messageBody.sdpMessageBody)) {

            v_SDP_Message_Step4or9or12 := v_Response.messageBody.sdpMessageBody;

          }

          f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

          if (f_MessageHeader_CheckRequire(v_MessageHeader, c_tag100rel)) {

            // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

            f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader);

          }

          f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer   @sic R5s130956 Change 2 sic@

        }

      [] t_Timer.timeout

        {

          f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer   @sic R5s130956 Change 2 sic@

        }

    }

    if (not isbound(v_SDP_Message_Step4or9or12)) {

      v_SDP_Message_Step12 := v_SDP_Message_RX;

    }

.......
  }

