
	Reason for change:
	1. Following the agreement of prose CR R5-142950, the description of step 2 states : “Steps 1-3 defined in annex C.22 with the exception that SS shall delay every response it sends to UE in such a way to cause half of the emergency registration time to expire after step 3”. The current TTCN implementation delays the 100 and 180 responses in such a way that the complete emergency registration time will have expired (or almost expired) by the time the 180 response is sent.This causes the test case to fail as the UE will typically initiate re-registration during one of the delay periods.
2. Following implementation of TTCN CR R5s140172, the altstep “a_TcpServer_AcceptConnection” will now handle a TCP_CLOSE_CNF as well as a TCP_ACCEPT_IND. After this message is handled, the TCP connection will have been cleared, which can cause the following statement in the calling altstep a_IP_IMS_TCP_ConnectionControl to throw a runtime error when it tries to perform a “valueof” operation on the variable v_UnprotectedConnectionByRef (which will have been reset to omit when the connectionwas closed)

	
	

	Summary of change:
	1. Calcualte the delay before sending the 100 and 183 responses with respect to half of the remaining Re-RegistrationTime (after INVITE has been received) .
2. Following execution of altstep a_TcpServer_AcceptConnection, check that the variable v_UnprotectedConnectionByRef is set to a value before performing the “valueof” operation.

	
	

	Consequences if not approved:
	A conformant UE may fail this test case.

Change 1
	Function name
	f_TC_19_5_7_IMS2()

	Reason for change
	Following the agreement of prose CR R5-142950, the description of step 2 states : “Steps 1-3 defined in annex C.22 with the exception that SS shall delay every response it sends to UE in such a way to cause the half of the emergency registration time to expire after step 3”. The current TTCN implementation delays the 100 and 180 responses in such a way that the complete emergency registration time will have expired (or almost expired) by the time the 180 response is sent.This causes the test case to fail as the UE will typically initiate re-registration during one of the delay periods.


	Summary of change
	v_DelayBeforeSending should be calculated with respect to half of the remaining ReRegistrationTime (after INVITE has been received) .

	TTCN module
	IMS_CC_EmergencyCallTestcases.ttcn


Before change

    function f_TC_19_5_7_IMS2() runs on IMS_PTC

  { /* IMS implementation for PDN2 - Emergency call */

…………..
    //@siclog "Step 2" siclog@

    t_RegExpireMax.start; // more than half of the expiration time shall elapse, used as well to calculate the elapsed time

    v_SDP_Message_Step1 := f_IMS_BuildSDP_AnnexC22_Step1();

    //@siclog "Step 1 of C.22" siclog@

    // Receive INVITE for an Emergency call

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A7, -, -, -, v_GeolocationInfoAvailable);

    v_ReRegistrationElapsedTime := t_RegExpireMax.read;

    t_RegExpireMax.stop; // not needed any more

    // If INVITE takes too long, t_RegExpireMax will time out and this will fail the TC

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_Message_Step1, v_GeolocationInfoAvailable);

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Message.media_list[0], omit);   // !!!! PROSE ISSUE: prose does not specify how to get the correct fmt !!!!

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    v_DelayBeforeSending := f_IMS_SetTimerToleranceMin((v_ReRegistrationTime - v_ReRegistrationElapsedTime) / 2.0); // calculate delay before sending before 100 Trying and 180 Ringing   @sic R5s140338 sic@

    // total time left before Reregistration divided equally for both messages to be sent after a delay

    //@siclog "Step 2 of C.22" siclog@

    // SS sends a 100 Trying provisional response.

    f_Delay(v_DelayBeforeSending); // delayed sending

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    //@siclog "Step 3 of C.22" siclog@

    // SS sends a 180 Ringing response.

    f_Delay(v_DelayBeforeSending); // delayed sending

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    //@siclog "Step 1-4 Parallel behaviour" siclog@

    // Parallel behaviour Steps 1 to 4 re-registration

    v_RegisterRequest := fl_IMS_EmergencyReRegistration();    // @sic R5s140338 sic@

    t_RegExpireMax.stop;

    f_IMS_TestBody_Set(false);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);

  }
After change

    function f_TC_19_5_7_IMS2() runs on IMS_PTC

  { /* IMS implementation for PDN2 - Emergency call */

…………
    //@siclog "Step 2" siclog@

    t_RegExpireMax.start; // more than half of the expiration time shall elapse, used as well to calculate the elapsed time

    v_SDP_Message_Step1 := f_IMS_BuildSDP_AnnexC22_Step1();

    //@siclog "Step 1 of C.22" siclog@

    // Receive INVITE for an Emergency call

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A7, -, -, -, v_GeolocationInfoAvailable);

    v_ReRegistrationElapsedTime := t_RegExpireMax.read;

    t_RegExpireMax.stop; // not needed any more

    // If INVITE takes too long, t_RegExpireMax will time out and this will fail the TC

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_Message_Step1, v_GeolocationInfoAvailable);

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Message.media_list[0], omit);   // !!!! PROSE ISSUE: prose does not specify how to get the correct fmt !!!!

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    v_DelayBeforeSending := f_IMS_SetTimerToleranceMin(((v_ReRegistrationTime / 2.0) - v_ReRegistrationElapsedTime) / 2.0); // calculate delay before sending before 100 Trying and 180 Ringing   @sic R5s140338 sic@

    // total time left before Reregistration divided equally for both messages to be sent after a delay

    //@siclog "Step 2 of C.22" siclog@

    // SS sends a 100 Trying provisional response.

    f_Delay(v_DelayBeforeSending); // delayed sending

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    //@siclog "Step 3 of C.22" siclog@

    // SS sends a 180 Ringing response.

    f_Delay(v_DelayBeforeSending); // delayed sending

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    //@siclog "Step 1-4 Parallel behaviour" siclog@

    // Parallel behaviour Steps 1 to 4 re-registration

    v_RegisterRequest := fl_IMS_EmergencyReRegistration();    // @sic R5s140338 sic@

    t_RegExpireMax.stop;

    f_IMS_TestBody_Set(false);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);

  }
Change 2
	Function name
	a_IP_IMS_TCP_ConnectionControl()

	Reason for change
	Following implementation of TTCN CR R5s140172, the altstep “a_TcpServer_AcceptConnection” will now handle a TCP_CLOSE_CNF as well as a TCP_ACCEPT_IND. After this message is handled, the TCP connection will have been cleared, which can cause the following statement in the calling altstep a_IP_IMS_TCP_ConnectionControl to throw a runtime error when it tries to perform a “valueof” operation on the variable v_UnprotectedConnectionByRef (which will have been reset to omit when the connectionwas closed)


	Summary of change
	Following execution of altstep a_TcpServer_AcceptConnection, check that the variable v_UnprotectedConnectionByRef is set to a value before performing the “valueof” operation.

	TTCN module
	IP_PTC_IMS_Handler.ttcn


Before change

    altstep a_IP_IMS_TCP_ConnectionControl(inout IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var IMS_TCP_Connections_Type v_TCP_Connections := p_ImsServer.TcpConnections;

    var template (omit) IP_Connection_Type v_UnprotectedConnectionByRef := v_TCP_Connections.Unprotected_uc_ps;

    var template (omit) IP_Connection_Type v_ProtectedConnectionByRef := v_TCP_Connections.Protected_uc_ps;

    var boolean v_Unprotected_uc_ps_IsConnected := isvalue(v_TCP_Connections.Unprotected_uc_ps);

    var boolean v_Unprotected_pc_us_IsConnected := isvalue(v_TCP_Connections.Unprotected_pc_us);

    var boolean v_Protected_uc_ps_IsConnected := isvalue(v_TCP_Connections.Protected_uc_ps);

    var boolean v_Protected_pc_us_IsConnected := isvalue(v_TCP_Connections.Protected_pc_us);

    var boolean v_ImsServerIsStarted := f_ImsServer_IsStarted(p_ImsServer);

    var boolean v_SecurityIsStarted := v_ImsServerIsStarted and fl_ImsServer_SecurityIsEstablished(p_ImsServer);

    var template (present) IP_AddrInfo_Type v_ServerAddrIPv4orIPv6 := (p_ImsServer.IPv4Addr, p_ImsServer.IPv6Addr);  // IPv4 or IPv6 address

    [v_ImsServerIsStarted] a_TcpServer_AcceptConnection(v_UnprotectedConnectionByRef, v_ServerAddrIPv4orIPv6, p_ImsServer.UnprotectedPorts.Port_ps)

      {

        p_ImsServer.TcpConnections.Unprotected_uc_ps := valueof(v_UnprotectedConnectionByRef);    // a_TcpServer_AcceptConnection 'returns' a valid connection i.e. v_UnprotectedConnectionByRef will never be omit
      }

    [v_SecurityIsStarted] a_TcpServer_AcceptConnection(v_ProtectedConnectionByRef, p_ImsServer.RegistrationInfo.NW_Address, p_ImsServer.RegistrationInfo.SecurityInfo.ProtectedPorts.Port_ps)

      {

        p_ImsServer.TcpConnections.Protected_uc_ps := valueof(v_ProtectedConnectionByRef);        // a_TcpServer_AcceptConnection 'returns' a valid connection i.e. v_UnprotectedConnectionByRef will never be omit

      }

    [v_Unprotected_uc_ps_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Unprotected_uc_ps)

      { // unprotected connection from UE to SS

        p_ImsServer.TcpConnections.Unprotected_uc_ps := omit;

      }

    [v_Unprotected_pc_us_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Unprotected_pc_us)

      { // unprotected connection from SS to UE

        p_ImsServer.TcpConnections.Unprotected_pc_us := omit;

      }

    [v_Protected_uc_ps_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Protected_uc_ps)

      { // protected connection from UE to SS

        p_ImsServer.TcpConnections.Protected_uc_ps := omit;

      }

    [v_Protected_pc_us_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Protected_pc_us)

      { // protected connection from SS to UE

        p_ImsServer.TcpConnections.Protected_pc_us := omit;

      }

  }
After change

    altstep a_IP_IMS_TCP_ConnectionControl(inout IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var IMS_TCP_Connections_Type v_TCP_Connections := p_ImsServer.TcpConnections;

    var template (omit) IP_Connection_Type v_UnprotectedConnectionByRef := v_TCP_Connections.Unprotected_uc_ps;

    var template (omit) IP_Connection_Type v_ProtectedConnectionByRef := v_TCP_Connections.Protected_uc_ps;

    var boolean v_Unprotected_uc_ps_IsConnected := isvalue(v_TCP_Connections.Unprotected_uc_ps);

    var boolean v_Unprotected_pc_us_IsConnected := isvalue(v_TCP_Connections.Unprotected_pc_us);

    var boolean v_Protected_uc_ps_IsConnected := isvalue(v_TCP_Connections.Protected_uc_ps);

    var boolean v_Protected_pc_us_IsConnected := isvalue(v_TCP_Connections.Protected_pc_us);

    var boolean v_ImsServerIsStarted := f_ImsServer_IsStarted(p_ImsServer);

    var boolean v_SecurityIsStarted := v_ImsServerIsStarted and fl_ImsServer_SecurityIsEstablished(p_ImsServer);

    var template (present) IP_AddrInfo_Type v_ServerAddrIPv4orIPv6 := (p_ImsServer.IPv4Addr, p_ImsServer.IPv6Addr);  // IPv4 or IPv6 address

    [v_ImsServerIsStarted] a_TcpServer_AcceptConnection(v_UnprotectedConnectionByRef, v_ServerAddrIPv4orIPv6, p_ImsServer.UnprotectedPorts.Port_ps)

      {

        if(isvalue(v_UnprotectedConnectionByRef)) {
        p_ImsServer.TcpConnections.Unprotected_uc_ps := valueof(v_UnprotectedConnectionByRef);    // a_TcpServer_AcceptConnection may ‘return’ a closed connection so v_UnprotectedConnectionByRef can be OMIT
       }
      }     

    [v_SecurityIsStarted] a_TcpServer_AcceptConnection(v_ProtectedConnectionByRef, p_ImsServer.RegistrationInfo.NW_Address, p_ImsServer.RegistrationInfo.SecurityInfo.ProtectedPorts.Port_ps)

      {

        p_ImsServer.TcpConnections.Protected_uc_ps := valueof(v_ProtectedConnectionByRef);        // a_TcpServer_AcceptConnection 'returns' a valid connection i.e. v_UnprotectedConnectionByRef will never be omit

      }

    [v_Unprotected_uc_ps_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Unprotected_uc_ps)

      { // unprotected connection from UE to SS

        p_ImsServer.TcpConnections.Unprotected_uc_ps := omit;

      }

    [v_Unprotected_pc_us_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Unprotected_pc_us)

      { // unprotected connection from SS to UE

        p_ImsServer.TcpConnections.Unprotected_pc_us := omit;

      }

    [v_Protected_uc_ps_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Protected_uc_ps)

      { // protected connection from UE to SS

        p_ImsServer.TcpConnections.Protected_uc_ps := omit;

      }

    [v_Protected_pc_us_IsConnected] a_TcpServer_CloseConnection(v_TCP_Connections.Protected_pc_us)

      { // protected connection from SS to UE

        p_ImsServer.TcpConnections.Protected_pc_us := omit;

      }

  }

